NAG Library Routine Document

C05AGF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

C05AGF locates a simple zero of a continuous function from a given starting value, using a binary search to locate an interval containing a zero of the function, then a combination of the methods of linear interpolation, extrapolation and bisection to locate the zero precisely.

2 Specification

```
SUBROUTINE C05AGF(X, H, EPS, ETA, F, A, B, IFAIL)
INTEGER             IFAIL
double precision    X, H, EPS, ETA, F, A, B
EXTERNAL            F
```

3 Description

C05AGF attempts to locate an interval $[a,b]$ containing a simple zero of the function $f(x)$ by a binary search starting from the initial point $x = \mathrm{X}$ and using repeated calls to C05AVF. If this search succeeds, then the zero is determined to a user-specified accuracy by a call to C05ADF. The specifications of routines C05ADF and C05AVF should be consulted for details of the methods used.

The approximation x to the zero α is determined so that at least one of the following criteria is satisfied:

(i) $|x - \alpha| \leq 4 \times \mathrm{EPS} \times \max(|x|, 1)$,

(ii) $|f(x)| \leq \mathrm{ETA}$.

4 References

Brent R P (1973) *Algorithms for Minimization Without Derivatives* Prentice–Hall

5 Parameters

1: X – ***double precision*** *Input/Output*

On entry: an initial approximation to the zero.

On exit: if IFAIL $= 0$ or 4, X is the final approximation to the zero. If IFAIL $= 3$, X is likely to be a pole of $f(x)$. Otherwise, X contains no useful information.

2: H – ***double precision*** *Input*

On entry: a step length for use in the binary search for an interval containing the zero. The maximum interval searched is $[\mathrm{X} - 256.0 \times \mathrm{H}, \mathrm{X} + 256.0 \times \mathrm{H}]$.

Constraint: H must be sufficiently large that $\mathrm{X} + \mathrm{H} \neq \mathrm{X}$ on the computer.

3: EPS – ***double precision*** *Input*

On entry: the termination tolerance on x (see Section 3).

Constraint: $\mathrm{EPS} > 0.0$.

4: ETA – ***double precision*** *Input*

On entry: a value such that if $|f(x)| \leq$ ETA, x is accepted as the zero. ETA may be specified as 0.0 (see Section 7).

5: F – ***double precision*** FUNCTION, supplied by the user. *External Procedure*

F must evaluate the function f whose zero is to be determined.

The specification of F is:

```
double precision FUNCTION F(XX)
double precision             XX
```

1: XX – ***double precision*** *Input*

On entry: the point at which the function must be evaluated.

F must be declared as EXTERNAL in the (sub)program from which C05AGF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

6: A – ***double precision*** *Output*
7: B – ***double precision*** *Output*

On exit: the lower and upper bounds respectively of the interval resulting from the binary search. If the zero is determined exactly such that $f(x) = 0.0$ or is determined so that $|f(x)| \leq$ ETA at any stage in the calculation, then on exit A = B = x.

8: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, either EPS ≤ 0.0, or X + H = X to machine accuracy (meaning that the search for an interval containing the zero cannot commence).

IFAIL = 2

An interval containing the zero could not be found. Increasing H and calling C05AGF again will increase the range searched for the zero. Decreasing H and calling C05AGF again will refine the mesh used in the search for the zero.

IFAIL = 3

A change in sign of $f(x)$ has been determined as occurring near the point defined by the final value of X. However, there is some evidence that this sign-change corresponds to a pole of $f(x)$.

IFAIL = 4

Too much accuracy has been requested in the computation; that is, the interval containing the zero has been reduced to one of relative length at most 2ϵ, where ϵ is the ***machine precision***, but the exit conditions described in Section 3 are not satisfied. It is unsafe to continue reducing the interval beyond this point, but the final value of X returned is an accurate approximation to the zero.

7 Accuracy

The levels of accuracy depend on the values of EPS and ETA. If full machine accuracy is required, they may be set very small, resulting in an exit with IFAIL = 4, although this may involve many more iterations than a lesser accuracy. You are recommended to set ETA = 0.0 and to use EPS to control the accuracy, unless you have considerable knowledge of the size of $f(x)$ for values of x near the zero.

8 Further Comments

The time taken by C05AGF depends primarily on the time spent evaluating F (see Section 5). The accuracy of the initial approximation X and the value of H will have a somewhat unpredictable effect on the timing.

If it is important to determine an interval of relative length less than 2 × EPS containing the zero, or if F is expensive to evaluate and the number of calls to F is to be restricted, then use of C05AVF followed by C05AZF is recommended. Use of this combination is also recommended when the structure of the problem to be solved does not permit a simple F to be written: the reverse communication facilities of these routines are more flexible than the direct communication of F required by C05AGF.

If the iteration terminates with successful exit and A = B = X there is no guarantee that the value returned in X corresponds to a simple zero and you should check whether it does.

One way to check this is to compute the derivative of f at the point X, preferably analytically, or, if this is not possible, numerically, perhaps by using a central difference estimate. If $f'(\mathrm{X}) = 0.0$, then X must correspond to a multiple zero of f rather than a simple zero.

9 Example

This example calculates an approximation to the zero of $x - e^{-x}$ using a tolerance of EPS = 1.0D−5 starting from X = 1.0 and using an initial search step H = 0.1.

9.1 Program Text

```
*     C05AGF Example Program Text
*     Mark 14 Revised. NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*     .. Local Scalars ..
      DOUBLE PRECISION A, B, EPS, ETA, H, X
      INTEGER          IFAIL
*     .. External Functions ..
      DOUBLE PRECISION F
      EXTERNAL         F
*     .. External Subroutines ..
      EXTERNAL         C05AGF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'C05AGF Example Program Results'
      X = 1.0D0
      H = 0.1D0
      EPS = 1.0D-5
      ETA = 0.0D0
      IFAIL = 1
*
      CALL C05AGF(X,H,EPS,ETA,F,A,B,IFAIL)
*
      WRITE (NOUT,*)
```

```
      IF (IFAIL.EQ.0) THEN
         WRITE (NOUT,99999) 'Root is ', X
         WRITE (NOUT,99998) 'Interval searched is [', A, ',', B, ']'
      ELSE
         WRITE (NOUT,99997) ' ** C05AGF returned with IFAIL = ', IFAIL
         IF (IFAIL.EQ.3 .OR. IFAIL.EQ.4) WRITE (NOUT,99999)
     +       'Final value = ', X
      END IF
*
99999 FORMAT (1X,A,F13.5)
99998 FORMAT (1X,A,F8.5,A,F8.5,A)
99997 FORMAT (1X,A,I5)
      END
*
      DOUBLE PRECISION FUNCTION F(X)
*     .. Scalar Arguments ..
      DOUBLE PRECISION X
*     .. Intrinsic Functions ..
      INTRINSIC        EXP
*     .. Executable Statements ..
      F = X - EXP(-X)
      RETURN
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 C05AGF Example Program Results

 Root is       0.56714
 Interval searched is [ 0.50000, 0.90000]
```
